unittest
{
	assert( [ "aab", "xyz" ].parse.expand.solve == "No" );		// 追加ケース
	assert( [ "xyz", "aab" ].parse.expand.solve == "No" );		// 追加ケース

	assert( [ "azzel", "apple" ].parse.expand.solve == "Yes" );
	assert( [ "chokudai", "redcoder" ].parse.expand.solve == "No" );
	assert( [ "abcdefghijklmnopqrstuvwxyz", "ibyhqfrekavclxjstdwgpzmonu" ].parse.expand.solve == "Yes" );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.string;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto ss = input.front.strip.to!( dchar[] );
	input.popFront;
	auto ts = input.front.strip.to!( dchar[] );
	return tuple( ss, ts );
}

auto solve( dchar[] ss, dchar[] ts )
{
	// 自分の中の文字に対応する位置の相手の文字を収集（つまり変換元⇔変換先の対応表を作る）
	dchar[ dchar ] sts;
	dchar[ dchar ] tss;
	foreach( s, t; lockstep( ss, ts ) )
	{
		if( s in sts && sts[ s ] != t ) return "No";		// Sの中に同じ文字が複数あって、それぞれに対応するTの文字が揃ってなかったら、もう無理
		sts[ s ] = t;
		if( t in tss && tss[ t ] != s ) return "No";		// Tの中に同じ文字が複数あって、それぞれに対応するSの文字が揃ってなかったら、もう無理
		tss[ t ] = s;
	}
	
	return "Yes";
}
